

Analyse, simulation d'un oscillateur COLPITTS

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet : [page d'accueil](#)

d'autres pdf, sur différents sujets : [liste des PDF](#)

Nous allons vérifier le principe de l'oscillateur et analyser par la simulation le fonctionnement d'un schéma à transistor.

1) Le principe

En électronique, parmi les différentes façons de réaliser un oscillateur, il existe la structure d'un amplificateur sur charge accordée, rebouclé :

L'amplificateur de tension a un coefficient A_v .

Le retour est pris sur la charge.

Étudions la chaîne de retour.

Hypothèse : l'entrée de l'amplificateur n'absorbe aucun courant

Posons Z l'association Z_1, Z_2, Z_3 :

$$Z = \frac{(Z_1 + Z_2)Z_3}{Z_1 + Z_2 + Z_3}$$




Associé à R , cela forme un pont diviseur de tension : $\frac{Z}{R + Z}$

$$\text{C'est à dire : } \frac{\frac{(Z_1 + Z_2)Z_3}{Z_1 + Z_2 + Z_3}}{R + \frac{(Z_1 + Z_2)Z_3}{Z_1 + Z_2 + Z_3}} = \frac{Z_2 Z_3}{R(Z_1 + Z_2 + Z_3) + (Z_1 + Z_2)Z_3}$$

La branche Z_1, Z_2 forme un pont diviseur de tension.

$$\text{On déduit } b = \frac{Z_2}{Z_2 + Z_1} \frac{Z_2 Z_3}{R(Z_1 + Z_2 + Z_3) + (Z_1 + Z_2)Z_3}$$

Parmi les différentes possibilités d'impédances Z_1, Z_2, Z_3 , l'oscillateur Colpitts est donné avec le schéma ci-contre :




Pour vérifier que ce montage forme un oscillateur, on exploite le formalisme des schémas blocs, et on se place dans des conditions d'instabilité. Pour ce faire, il nous faut disposer de la réponse harmonique de la boucle ouverte, Ab .

En régime harmonique, nous avons : $Z_1 = \frac{1}{jC_1 \omega}$ $Z_2 = \frac{1}{jC_2 \omega}$ $Z_3 = jL\omega$

$$\text{On déduit } b(j\omega) = \frac{C_1}{C_1 + C_2} \frac{1}{R \left(\frac{1}{L j\omega} + C_e j\omega \right) + 1} \quad \text{où } C_e = \frac{C_1 C_2}{C_1 + C_2}$$

La théorie des systèmes bouclés donne cette structure oscillante en régime sinusoïdal pour :

module de $A(j\omega)b(j\omega) = 1$ et $Arg\{A(j\omega)b(j\omega)\} = 0$



Dans notre cas : $A(j\omega) = A_v$ soit réel. Il faut donc phase de $b(j\omega) = 0$, soit $b(j\omega)$ réel également.

Ce qui signifie $\frac{1}{L j\omega} + C_e j\omega = 0$. Ceci est obtenu pour la pulsation $\omega_0 = \frac{1}{\sqrt{L C_e}}$.

À cette pulsation ω_0 , le **module de Ab doit être = 1**, soit $A_v \frac{C_1}{C_1 + C_2} = 1$ c'est-à-dire $A_v = 1 + \frac{C_2}{C_1}$.

Si le module est > 1 , le système bouclé sera instable, et présentera une oscillation d'amplitude divergente. En pratique, l'amplitude sera bornée à cause de non linéarité apportée par l'amplificateur (saturation par exemple), et le régime ne sera pas sinusoïdal. On parle de régime quasi sinusoïdal si la forme d'onde n'est pas trop éloignée d'une sinusoïde.

Une solution pour réaliser l'amplificateur de coefficient A_v est d'exploiter un amplificateur opérationnel :

comme $A_v = 1 + \frac{R_2}{R_1}$, il faut, pour osciller, $\frac{R_2}{R_1} \geq \frac{C_2}{C_1}$




Schéma à A. Op.

Si on veut travailler en fréquences élevées, il est préférable d'exploiter un montage amplificateur à transistor. On élimine l'émetteur commun (qui est inverseur), le collecteur commun (qui n'amplifie pas les tensions), et on exploite le montage base commune.

Sans aucune optimisation, le montage serait celui-ci-contre :

R_{B1} R_{B2} forme la polarisation.

C_{dec} ramène une masse dynamique à la base.

La sortie est sur le collecteur. L'entrée est sur l'émetteur.

C_{Lout} coupe la composante continue.

C_{Lin} est placé usuellement, mais dans ce montage, n'a pas de rôle, car la composante continue est déjà bloquée.




Schéma de principe à transistor

On a : $R_{B1} = 22 \text{ k}\Omega$; $R_{B2} = 150 \text{ k}\Omega$; $R_C = 5,6 \text{ k}\Omega$; $R_E = 1040 \Omega$
Tension d'alimentation : 15 V. Transistor Q2N2222.

Prédétermination de la **polarisation** : on cherche I_C , V_{CE} , statique.

On pose a priori $I_P \gg I_B$, $V_{BE} = 0,65 \text{ V}$. On déduit : $I_P = 15 \text{ V} / 172\text{k} = 87 \mu\text{A}$.

Le pont diviseur donne $V_B = 15 \times 22\text{k} / 172\text{k} = 1,92 \text{ V}$.

Cela donne un potentiel d'émetteur de $1,92 \text{ V} - 0,65 \text{ V} = 1,27 \text{ V}$.

Par loi d'Ohm, on calcule un courant d'émetteur de $1,27 / 1040 = 1,22 \text{ mA} = I_C$.

Il s'ensuit $V_C = 15 - (5,6\text{k} \times 1,22\text{m}) = 8,17 \text{ V}$, et donc $V_{CE} = V_C - V_E = 8,17 - 1,27 = 6,9 \text{ V}$.

La répartition des potentiels est satisfaisante.

Remarque : si on considère que β est environ 150, on a $I_B \approx 8 \mu\text{A}$. L'hypothèse $I_P \gg I_B$ est vérifiée.

Le calcul précédent reste valable.

Déterminons l'**amplification**.

Dans un base commune, l'amplification est R_C / R_E .

On a $R_C = 5600\Omega$, $R_E = 1040\Omega$. L'amplification attendue à vide de ce montage est donc 5,38.

2) Simulations

Vérifions les différents points qui viennent d'être étudiés :

a) La chaine de retour


Posons des repères C, E, S, sur le schéma.

Prenons $C1 = C2 = 30 \text{ pF}$,
 $R = 47 \text{ } \Omega$, $L = 220 \text{ nH}$.

On déduit :

$Ce = 15 \text{ pF}$, et $f_0 = 87,6 \text{ MHz}$.


Traçons la réponse harmonique de la chaine de retour b, ici $V(E)/V(C)$.



```

chaîne de retour de l'oscillateur Colpitts
* fichier pontdiviseur.cir
* circuit
Vin C 0 AC 1
R C S 47
C1 S E 30p
C2 E 0 30p
Rcharge E 0 1Meg
L S 0 220n
.AC DEC 100 1MEG 10G
.probe
.end
    
```

Netlist, prête à simuler



Réponse harmonique du pont diviseur

Interprétation :

On remarque une résonance à 87,6 MHz. A cette fréquence la phase vaut 0° . Le gain vaut alors -6 dB .


Cette valeur est conforme à la théorie, puisqu'à cette fréquence, b est égal à $\frac{C1}{C1+C2}$, c'est-à-dire 0,5.

b) Le système bouclé avec un amplificateur parfait et saturations

Vérifions dans un deuxième temps le fonctionnement en oscillateur, avec un amplificateur parfait. Pour avoir oscillation, l'amplification doit être supérieure à 2. Fixons arbitrairement 2,5.

On exploite une source de tension commandée par une tension, disponible dans la bibliothèque de Pspice.

On lui ajoute un écrêteur à 0,5 V et 14,5 V au nœud C pour modéliser les limites basse et haute d'un circuit alimenté en 15 V.




**Structure Colpitts
 L'ampli est à $Av = 2,5$, et sa tension de sortie est limitée**

```

Oscillateur colpitts a source de tension
* fichier colpitts_parfait.cir
* circuit
Ec C 0 table {2.5*V(E)} (0.5 0.5 14.5 14.5)
R C S 47
C1 S E 30p
C2 E 0 30p
Rcharge E 0 1Meg
L S 0 220n
.IC V(E)=2V
.TRAN 0.1ns 150n 0n 0.1ns
.probe
.end
    
```

Netlist, prête à simuler



Réponse temporelle : oscillation

Interprétation :

En simulation, les conditions initiales lancent l'oscillation. La fréquence obtenue est de 87,6 MHz, ce qui était attendu. En C, l'amplitude diverge mais est écrêtée à 0,5 V et 14,5 V, par les butées ajoutées. En S, le signal n'a pas l'allure rectangulaire comme au nœud C, mais n'est pas sinusoïdal non plus !

c) Le système, non bouclé, avec un amplificateur à transistor

Vérifions maintenant l'étage amplificateur, réalisé par un transistor 2N2222 monté en base commune, alimenté en 15 V.

La polarisation a été prédéterminée : I_C attendu voisin de 1,22 mA.




Schéma pour vérifier la polarisation

```
Oscillateur colpitts
* fichier colpitts.cir
.LIB EVAL.LIB
* circuit
* chaîne directe (A)
VCC A 0 DC=15
RB2 A B 150k
RB1 B 0 22k
Cde B 0 100n
Q1 C B E Q2N2222 ; C B E
RE E 0 1040
RC A C 5.6k
.OP
Début de la netlist
```

Par la directive .OP, vérifions tout d'abord la **polarisation** : Il nous faut consulter le fichier de sortie.

Interprétation :

On remarque V_{BE} direct, V_{BC} inverse, ce qui signifie le mode normal de polarisation.

Le courant de collecteur est de 1,09 mA, valeur usuelle pour ce type de transistor discret, et proche de la prédétermination faite « à la main ».

On peut lire également $g_m = 0,0421$. On déduit la résistance d'entrée de ce montage base commune $1/g_m = 23,75 \Omega$. Cette faible valeur est le propre du montage base commune.

| | |
|----------|-----------|
| NAME | Q1 |
| MODEL | Q2N2222 |
| IB | 6.75E-06 |
| IC | 1.09E-03 |
| VBE | 6.46E-01 |
| VBC | -7.09E+00 |
| VCE | 7.74E+00 |
| BETA DC | 1.62E+02 |
| GM | 4.21E-02 |
| RPI | 4.25E+03 |
| RX | 1.00E+01 |
| RO | 7.43E+04 |
| CBE | 5.37E-11 |
| CBC | 3.28E-12 |
| CJS | 0.00E+00 |
| BETA AC | 1.79E+02 |
| CBX/CBX2 | 0.00E+00 |
| FT/FT2 | 1.18E+08 |

Extrait du fichier de sortie

Vérifions ensuite les conditions d'oscillation, en analysant la fonction **Ab**, et ce, le circuit n'étant pas rebouclé.

Pour ce faire, on construit un circuit formé de A cascadé avec b.

Notons que la branche de retour b est chargée par une résistance de 23 Ω, pour représenter sommairement l'entrée du montage base commune (on aurait pu placer un autre transistor câblé comme l'amplificateur pour mieux représenter son comportement)

On a appelé E1 le nœud en sortie de la branche de retour.

La transmittance $E1/Vin$ est donc le produit **Ab**.




Schéma testé




Schéma bloc équivalent

```


Oscillateur colpitts
* fichier colpitts.cir
.LIB EVAL.LIB
* circuit
* chaîne directe (A)
VCC A 0 DC=15
RB2 A B 150k
RB1 B 0 22k
Cde B 0 100n
Q1 C B E Q2N2222 ; C B E
RE E 0 1040
RC A C 5.6k
.OP

* chaîne de retour (b)
CLout C S 100n
C1 S E1 30p
C2 E1 0 30p
RinBC E1 0 23
L S 0 220n ; self

* pour tester A b
Vin in 0 AC 1
CLin in E 100n
.AC DEC 100 1 100G

* pour verifier oscillation
*.IC V(E)=2V
*.TRAN 0.1ns 10u 0n 0.1ns
.probe
.end

Netlist, prête à simuler
    
```



Réponse harmonique de la chaîne Ab, chargée par 23 Ω

Trace supérieure : $V(E1)$. Comme $V(Vin)=1$, cette trace représente également l'amplification.

Trace inférieure : phase.

Interprétation :

Cette réponse est très... biscornue ! D'où l'intérêt de la simulation, car l'équation théorique est très lourde. On peut lire qu'à la fréquence pour laquelle le déphasage = 0 (traversée de -360 °), le module vaut 2,24 et est donc supérieur à 1 : **les conditions d'oscillations sont réalisées.**

d) Le système bouclé


Il ne reste plus qu'à reboucler, et tester le montage en temporel.

Remarques : netlist à réadapter :


```
* chaîne de retour (b)
CLout C S 100n
C1 S E 30p
C2 E 0 30p
*RinBC E 0 23
L S 0 220n ; self

* pour tester A b
*Vin in 0 AC 1
*CLin in E 100n
*.AC DEC 100 1 100G

* pour vérifier oscillation
.IC V(E)=2V
.TRAN 0.1ns 10u 0n 0.1ns
.probe
.end
```



On boucle les 2 étages



En V(S) on a bien l'oscillation. La fréquence est de 74,3 MHz.

e) Schéma final

Il existe le schéma usuel pour l'oscillateur Colpitts à transistor.

L'inductance L est placée entre le collecteur et l'alimentation, ce qui est équivalent en dynamique. Par ce fait, RC est inutile.

La résistance R est retirée.

La branche de retour reste le point commun C1 C2 et l'émetteur.

La maille RB1 RB2 est identique, donc le potentiel de base est inchangé.

Le point de polarisation est légèrement décalé, car VCE est augmenté.





Schéma final.

La sortie est au collecteur.

```
Oscillateur colpitts
* fichier colpitts_2.cir
.LIB EVAL.LIB
* circuit
* chaîne directe (A)
VCC A 0 DC=15
RB2 A B 150k
RB1 B 0 22k
Cde B 0 100n
Q1 C B E Q2N2222 ; C B E
RE E 0 1040
L A C 220n ; forme également
* la chaîne de retour (b)
C1 C E 30p
C2 E 0 30p
*.OP
.IC V(E)=2V
.TRAN 10p 1u 0.3u 10p
.probe
.end
```

Netlist prête à simuler



Zoom (50 ns) sur les potentiels de collecteur et d'émetteur en régime établi

Interprétation :


La fréquence théorique est donnée par $f_0 = \frac{1}{2\pi\sqrt{L C_e}} = 87,6 \text{ MHz}$.

Cette dernière simulation donne une fréquence de 72,9 MHz. Cet écart de 15 % peut s'expliquer par l'apport des capacités de jonction : C_{BE} s'ajoute à C_2 et C_{BC} s'ajoute à (C_1 série C_2).

Néanmoins, rappelons que la théorie a été établie avec des hypothèses de systèmes linéaires, ce qui peut être validé dans le domaine petits signaux. Ici, les formes d'ondes montrent qu'on est en grands signaux, avec des non linéarités, comme la saturation du transistor. De plus, la branche de retour débite dans l'impédance d'entrée de la base commune, qui n'est pas infinie, comme l'amplificateur parfait.


Taux de distorsion

Un signal sinusoïdal présente un spectre formé d'une raie (en représentation unilatérale). Un signal qui quitte un aspect sinusoïdal aura un spectre formé de raies. Pour quantifier la « qualité » d'une sinusoïde, on exprime le taux de distorsion, qui exploite les amplitudes des harmoniques.



Une expression du taux de distorsion est : $d = \frac{\sqrt{A_1^2 + A_2^2 + A_3^2 + \dots}}{A_0}$.

Par la FFT, on peut afficher les amplitudes des harmoniques du signal issu de l'oscillateur Colpitts :




Coordonnées de chaque raie (hormis la fréquence 0 Hz) :

| | | | | |
|-----------|-----------|-----------|-----------|------------|
| 72,99 MHz | 145,9 MHz | 218,9 MHz | 291,9 MHz | 364,84 MHz |
| 14,29 V | 1,31 V | 0,622 V | 0,296 V | 0,1033 V |

On déduit le taux de distorsion : $\frac{\sqrt{1,31^2 + 0,622^2 + 0,296^2 + 0,1033^2 \dots}}{14,29} \approx 10,4\%$.

On peut également observer le **courant dans l'inductance** :



*La forme d'onde du courant dans l'inductance est proche de la sinusoïde.
Son taux de distorsion n'est que de 5%*

3) Conclusion

Comme de nombreux oscillateurs, le Colpitts est articulé autour d'un système bouclé instable.

Une approche purement électronique aurait consisté à exploiter le schéma dynamique du transistor (source de courant i_c commandée par une tension v_{be}), associé à un circuit résonant.

L'approche donnée dans ce document montre un angle de vue d'automatique, en exploitant le formalisme des schémas blocs. C'est pour cela qu'il a été décrit par une association amplificateur de tension et branche de retour, puis simplification. Malgré les simplifications de calcul données au départ, on a un fonctionnement du schéma « vrai » à transistor très proche de la prédétermination.

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

| n° | titre | lien présentation | lien direct article |
|----|--|------------------------------|------------------------------|
| | Guide d'installation et d'emploi simplifié | présentation | document PDF |
| 44 | Exemples basiques et des exercices... | présentation | document PDF |
| 45 | Un exemple de circuit passif | présentation | document PDF |
| 46 | Un oscillateur Colpitts | présentation | document PDF |
| 47 | Compensation en fréquence des amplificateurs opérationnels | présentation | document PDF |
| 48 | Un amplificateur à transistors bipolaires | présentation | document PDF |
| 49 | Une bascule D Flip Flop CMOS | présentation | document PDF |
| 50 | Une porte XOR à transistors MOS | présentation | document PDF |
| 51 | Un VCO à 12 transistors MOS | présentation | document PDF |
| 52 | Une PLL à moins de 20 transistors MOS | présentation | document PDF |
| 53 | Un oscillateur à résistance négative | présentation | document PDF |
| 54 | Une charge électronique | présentation | document PDF |
| 55 | Un amplificateur en classe C | présentation | document PDF |
| 56 | Le monostable 74 123 | présentation | document PDF |
| 57 | Un amplificateur en classe D | présentation | document PDF |
| 58 | Le transformateur en linéaire | présentation | document PDF |
| 59 | La loi d'ohm thermique | présentation | document PDF |
| 60 | Le transformateur en non linéaire | présentation | document PDF |
| 61 | Robustesse d'un oscillateur en anneau | présentation | document PDF |
| 62 | Une alimentation stabilisée | présentation | document PDF |
| 63 | Modélisation d'un haut-parleur | présentation | document PDF |
| 64 | Un synthétiseur de fréquence | présentation | document PDF |
| 65 | Un ampli audio de Sparkfun | présentation | document PDF |
| 66 | Simulation logique et analogique | présentation | document PDF |
| 67 | Un oscillateur à relaxation | présentation | document PDF |
| 68 | Lecteur de TAG RFID 125 kHz | présentation | document PDF |
| 69 | Diagramme de l'œil avec Pspice | présentation | document PDF |
| 70 | Un amplificateur hautes fréquences | présentation | document PDF |
| 71 | Une bizarrerie enfin expliquée... | présentation | document PDF |
| 72 | Comprendre le paramétrage de la FFT | présentation | document PDF |
| 73 | La relation de Bennett | présentation | document PDF |
| 74 | Simuler un circuit à plus de 20 transistors avec PSpice Eval | présentation | document PDF |
| 75 | Une horloge biphase sans recouvrement | présentation | document PDF |
| 76 | Quelques simulations sur la diode | présentation | document PDF |
| 77 | Un ampli classe A, avec transformateur de sortie | présentation | document PDF |
| 78 | Des stimuli pour PSpice | présentation | document PDF |
| 79 | Simuler le TL431 : zener ajustable | présentation | document PDF |
| 80 | Un ADC flash | présentation | document PDF |
| 81 | Une chaîne d'acquisition : S&H, ADC, DAC | présentation | document PDF |
| 82 | Un amplificateur 50 MHz | présentation | document PDF |
| 83 | Un dérivateur non inverseur | présentation | document PDF |
| 84 | Un amplificateur bipolaire avec push pull CMOS | présentation | document PDF |
| 85 | Rôle des répéteurs logiques dans un circuit intégré | présentation | document PDF |
| 86 | Un driver logique CMOS pour charge 50 ohms | présentation | document PDF |
| 87 | Des triggers de Schmitt et des applications | présentation | document PDF |
| 88 | Un filtre gaussien analogique | présentation | document PDF |
| 89 | Un générateur de bruit rose | présentation | document PDF |

| | | | |
|--|--|------------------------------|------------------------------|
| 90 | Un anémomètre à fil chaud : simulation comportementale | présentation | document PDF |
| 91 | Un oscillateur à pont de Wien stabilisé par CTN | présentation | document PDF |
| 92 | L'emballage thermique d'une diode | présentation | document PDF |
| 93 | Les puissances dans un amplificateur | présentation | document PDF |
| 94 | Asservissement de puissance dans une résistance | présentation | document PDF |
| 95 | Asservissement de la puissance émise par une antenne radio | présentation | document PDF |
| 96 | Un driver de LED de puissance | présentation | document PDF |
| 97 | Exploiter Pspice pour simuler des filtres numériques | présentation | document PDF |
| 98 | Un filtre en cosinus surélevé avec Pspice | présentation | document PDF |
| 99 | Effet de la température sur un amplificateur en classe A | présentation | document PDF |
| 100 | Un amplificateur à transistors JFET et bipolaires | présentation | document PDF |
| | | | |
| | | | |
| | | | |
| | | | |
| | | | |
| | | | |
| | | | |
| Supplément, hors article : | | | |
| mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf | | | |

[retour à l'écran d'accueil de ce site](#)